


UNIVERSIDADE DE SÃO PAULO

Roletas animadas

Ton Maran

Nº 47

NOTAS DIDÁTICAS



Instituto de Ciências Matemáticas de São Carlos


UNIVERSIDADE DE SÃO PAULO
Instituto de Ciências Matemáticas e de Computação
ISSN 0103-2585

Roletas animadas

Ton Maran

Nº 47

NOTAS DIDÁTICAS



São Carlos – SP
Nov./2000

Roletas animadas

Ton Maran

Departamento de Matemática
ICMC/USP
Caixa Postal 668
13560-970, São Carlos, SP, Brazil

e-mail: ton@icmc.sc.usp.br

Roletas

1. INTRODUÇÃO

No presente texto o leitor encontra uma introdução às curvas denominadas *roletas*, em particular as *trocóides*. O tópico é bastante visual, criando um bom pretexto para a utilização do programa de manipulação algébrica MAPLE. No final do texto encontram-se programas, criados pelo autor, onde são utilizadas ferramentas MAPLE para animação. Qualquer comentário pode ser enviado para ton@icmc.sc.usp.br.

2. DEFINIÇÃO

Sejam C_1 e C_2 duas curvas. Seja $P \in C_1$ um ponto fixado. Quando a curva C_1 rola sobre a curva C_2 , sem deslizar, o ponto P traça uma curva denominada *roleta*.

3. CICLÓIDES





Um exemplo famoso de roleta é aquele obtido quando C_1 e C_2 são respectivamente um círculo e uma reta. Neste caso, a curva traçada por um ponto fixo de C_1 , quando esta curva rola sobre C_2 , é chamada *ciclóide*.




4. TROCÓIDES

Quando C_1 e C_2 são ambos círculos a roleta recebe o nome de *trocóide*.

4.1. Caso 1 :- C_1 interna a C_2 . Neste caso a curva traçada é chamada *hipotrocóide*.




4.2. Caso 2 :- C_1 externa a C_2 . Neste caso a curva traçada é chamada *epitrocóide*.



5. EQUACIONAMENTO

5.1. Ciclóide. Consideremos o sistema de coordenadas cartesianas $\{O, x, y\}$ onde a origem O é o ponto inicial da ciclóide, o eixo Ox é a reta C_2 e portanto, o eixo Oy passa pelo centro de C_1 na posição inicial.




Quando o círculo C_1 rola sobre a reta C_2 de um ângulo α , seu centro percorre a distância α vezes o raio de C_1 . Sendo r o raio de C_1 , as coordenadas do ponto $P = (x, y) \in C_1$, após o giro do ângulo α , são:

$$\begin{aligned} x &= \alpha r - r \sin(\alpha) \\ y &= r - r \cos(\alpha) \end{aligned}$$

Assim, $t \rightarrow (tr - r \sin(t), r - r \cos(t))$ é uma parametrização da ciclóide, onde t é o ângulo formado pelo raio de C_1 com a vertical, medido no sentido horário.

5.2. Hipotrocóide. Sejam C_1 e C_2 respectivamente os círculos móvel (menor) e fixo (maior) de raios r e R respectivamente. Consideremos o sistema de coordenadas cartesianas $\{O, x, y\}$ onde a origem O é o centro de C_2 , como na figura abaixo:




O arco percorrido pelo círculo menor tem comprimento θR . Assim, o ângulo α pelo qual o círculo menor rolou sobre o círculo maior verifica a igualdade: $\alpha r = \theta R$. Logo, o ponto $P = (x, y)$ traça uma curva que verifica:

$$\begin{aligned} x &= (R - r) \cos(\theta) + r \cos((R/r - 1)\theta) \\ y &= (R - r) \sin(\theta) - r \sin((R/r - 1)\theta) \end{aligned}$$

Assim, $t \rightarrow ((R - r) \cos(t) + r \cos((R/r - 1)t), (R - r) \sin(t) - r \sin((R/r - 1)t))$ é uma parametrização da hipotrocóide, onde o círculo maior tem raio R e o menor tem raio r .

5.3. Epitrocóide.



Analogamente ao caso anterior, as coordenadas de um ponto $P = (x, y)$ da curva verificam:

$$\begin{aligned} x &= (R + r) \cos(\theta) + r \cos(\pi - (R/r + 1)\theta) \\ y &= (R + r) \sin(\theta) - r \sin(\pi - (R/r + 1)\theta) \end{aligned}$$

Assim, $t \rightarrow ((R+r) \cos(t) + r \cos(\pi - (R/r+1)t), (R+r) \sin(t) - r \sin(\pi - (R/r+1)t))$ é uma parametrização da epitrocóide, onde o círculo maior tem raio R e o menor tem raio r .

6. ALGUMAS PROPRIEDADES

6.1. Ciclóides. Vamos listar apenas algumas propriedades dessa roleta. As propriedades das ciclóides poderiam preencher as páginas de longos livros

1) A ciclóide é a solução de um problema clássico denominado *braquistócrona*. Este é o problema de determinar o formato de um escorregador que leva uma partícula, sob a ação da gravidade e sem atrito, em tempo mínimo do extremo superior ao extremo inferior. A solução é um arco de ciclóide. Além disso, de qualquer ponto intermediário deste escorregador, a partícula ainda levará o mesmo tempo (mínimo) para chegar até o extremo inferior. Tal propriedade é denominada *isocronismo*. Os relógios de pêndulo, que não devem depender do tamanho da oscilação, percorrem arcos de ciclóide. O primeiro a patentear um relógio de pêndulo, Christian Huygens, foi quem em 1656 deduziu a equação da ciclóide e muitas de suas propriedades.

2) Note que o centro do círculo C_1 percorre uma reta paralela à reta C_2 a uma altura igual ao raio r de C_1 . Daí o conforto de se movimentar num automóvel com rodas circulares sobre uma estrada plana! Também é possível obter um movimento semelhante se C_1 fosse, por exemplo um quadrado. Isto é, um carro de rodas quadradas poderia ainda fornecer um certo conforto, ao ponto de os seus acupantes nem notarem a diferença, caso a estrada, isto é, a curva C_2 sobre a qual rola C_1 fosse adequada.

Estudos como este do automóvel de roda quadrada, e mais geral ainda, rodas poligonais, são encontrados no interessante artigo de Leon Hall e Stan Wagon, *Roads and wheels*. Math. Mag. 65 (1992), no. 5, 283–301.

No caso de um automóvel de rodas quadradas, seu centro ainda se mantém numa reta durante o movimento se o quadrado estiver rolando sobre uma adequada ciclóide. Portanto, para seu conforto, em estradas cicloides só mesmo um carro de rodas quadradas!

3) Velocidade

Suponhamos que o círculo rola com velocidade angular constante descrevendo a ciclóide. No entanto, o ponto percorrendo a ciclóide tem velocidade bastante variável. Note na figura abaixo, à esquerda o quanto o ponto percorre depois que o círculo rola $\frac{\pi}{2}$ e à direita o quanto foi percorrido depois de rolar π .

De fato, a ciclóide parametrizada por $t \rightarrow (tr - r \sin(t), r - r \cos(t))$ tem (vetor) velocidade dada por $t \rightarrow (r - r \cos(t), r \sin(t))$, cujo módulo é $\sqrt{2r^2 - 2r^2 \cos(t)} = \sqrt{2r^2(1 - \cos(t))} = \sqrt{2r^2 2\sin^2(\frac{t}{2})} = 2r \sin(\frac{t}{2})$. Logo, quando t varia de 0 a π , a velocidade varia em módulo proporcional à variação do seno de 0 a $\frac{\pi}{2}$.




Além disso, o ângulo θ formado pelo vetor velocidade e a horizontal se relaciona com o ângulo t usado na parametrização da ciclóide da seguinte forma: $\cos(\theta) = \sin(\frac{t}{2})$, ou seja, $\theta = \frac{\pi-t}{2}$ (exercício).

4) Área

A área limitada entre a ciclóide e a reta sobre a qual o círculo rola, depois de uma revolução completa do círculo, é 3 vezes a área do círculo. Assim, na figura abaixo, o círculo divide a região em três partes de áreas iguais.

De fato, sendo $x = tr - r \sin(t)$ e $y = r - r \cos(t)$ então $dx = (r - r \cos(t))dt$. Entre duas cúspides da ciclóide, isto é, para x variando de 0 a $2\pi r$, temos que t varia de 0 a 2π . Logo, a área A sob a ciclóide entre duas cúspides é dada por:

$$A = \int_0^{2\pi r} y dx = \int_0^{2\pi} (r - r \cos(t))(r - r \cos(t)) dt = r^2 \int_0^{2\pi} (1 - \cos(t))^2 dt = 3\pi r^2.$$



6.2. Trocóides. A relação entre os raios dos círculos determinam a geometria das curvas traçadas. Se (R, r) são inteiros primos entre si, então a trocóide é uma curva fechada com R pontos cuspidais e fechará depois que o círculo móvel percorrer r vezes o círculo fixo. Quando (R, r) não forem primos entre si, reduzimos a fração $\frac{R}{r}$ e o mesmo se verifica. No caso particular, quando $\frac{R}{r} = 2$, a hipotrocóide correspondente é um segmento de reta (diâmetro do círculo fixo). Temos neste caso a transformação de movimento circular em movimento retilíneo, propriedade de interesse em mecanismos. O leitor é convidado a verificar essas propriedades com a utilização dos programas Maple, e a seguir, demonstrar tais afirmações utilizando as parametrizações das curvas. Finalmente, quando $\frac{R}{r}$ é irracional, a trocóide nunca se fecha.


7. PERTURBAÇÕES

Uma pequena perturbação no ponto de traçado, isto é, se ao invés de um ponto no bordo do círculo em movimento tomarmos como ponto de traçado um ponto exterior (ou interior) desse círculo, as cúspides que aparecem nas roletas se desdobrarão.

No caso das ciclóides, a parametrização torna-se:


$$\begin{aligned}x(t) &= rt - (r + a)\sin(t) \\y(t) &= r - (r + a)\cos(t)\end{aligned}$$

onde a é a medida do quanto o ponto de traçado se afasta do bordo de C_1 .



A curva traçada pela ciclóide quando $a > 0$ é aquela traçada por um ponto da roda de um trem. A porção a é o quanto a roda do trem está encaixada no trilho. Neste caso temos uma situação paradoxal, pois enquanto o trem move-se para frente, existem pontos dele que se movem para trás !

No caso das trocóides



a parametrização de uma perturbação é

$$\begin{aligned}x(t) &= (R + r) \cos(t) + (r + a) \cos(\pi - (R/r + 1)t) \\y(t) &= (R + r) \sin(t) - (r + a) \sin(\pi - (R/r + 1)t)\end{aligned}$$

Uma curva familiar pode ser obtida como uma hipotrocóide perturbada no caso de $R = 2r$, que curva é essa ?

Animação de traçado de curvas trocoidais

(não é necessário copiar os comentários, copie apenas os comandos que são indicados por >)

Hipotrocóides

r - raio do círculo móvel

> r:=2:

R - raio do círculo fixo

> R:=5:

a - distância do ponto de traçado ao bordo do círculo móvel

se $a>0$ o ponto de traçado está fora do círculo móvel, se $a<0$ está dentro e se $a=0$ o ponto está sobre a circunferência

> a:=-1:

n - número de subdivisão do ângulo percorrido

> n:=50:

N - número de voltas que o círculo móvel faz

> N:=5:

Equações: $(x,y)=(x(u),y(u))$ parametrização da curva a ser traçada

> x:=(R-r)*cos(u)+(r+a)*cos((R/r-1)*u):

> y:=(R-r)*sin(u)-(r+a)*sin((R/r-1)*u):

with(plots) lança o pacote contendo diversos tipos especiais de plots, exemplo **Polygonplot** e **display**, que serão usados abaixo

> with(plots):

início da rotina para a confecção de cada quadro a ser exibido, um quadro para cada i ($i = 1, \dots, n$)

> for i from 1 to n do

te - ângulo total percorrido pelo círculo móvel

U - subdivisão do ângulo te utilizado na confecção de cada quadro a ser exibido

> te:=2*N*Pi; U:=(te*i)/n:

p.i é a seqüência de plots que geram a curva

g.i é a seqüência de plots que apresentam o movimento do círculo móvel

> p.i := plot ({[r*cos(t) + (R-r)*cos(U), r*sin(t) + (R-r)*sin(U), t = 0..2*Pi], [R*cos(v), R*sin(v), v = 0..2*Pi]}, scaling = constrained, color=blue):

> g.i := plot ([x,y,u=0..U], scaling = constrained, color=red):

q1.i, q2.i e q3.i são seqüências de plots de segmentos radiais

isto é apenas um detalhe cosmético que auxilia na visualização do movimento do círculo móvel

note que q1.i varia de tamanho conforme o valor de a (distância do ponto de traçado ao bordo do círculo móvel)

```
> q1.i:= polygonplot ([[((R - r)*cos(U), (R - r)*sin(U)),
[(R - r)*cos(U) + (r + a)*cos((R/r - 1)*U), (R - r)*sin(U) - (r + a) * sin(
(R/r - 1)*U)]], scaling = constrained, axes = none, thickness = 2):
> q2.i:= polygonplot ([[((R - r)*cos(U), (R - r)*sin(U)),
[(R - r)*cos(U) + (r)*cos((R/r - 1)*U + 2/3*Pi), (R - r)*sin(U) - (r) * sin(
(R/r - 1)*U + 2/3*Pi)]], scaling = constrained, axes = none):
> q3.i:= polygonplot ([[((R - r)*cos(U), (R - r)*sin(U)),
[(R - r)*cos(U) + (r)*cos((R/r - 1)*U + 4/3*Pi), (R - r)*sin(U) - (r) * sin(
(R/r - 1)*U + 4/3*Pi)]], scaling = constrained, axes = none):
```

```
> od:
```

Rotina para exibição dos quadros

p=curva, g=círculo móvel, q1, q2 e q3= segmentos radiais

```
> T:=NULL:
```

```
> for s from 1 to n do
```

```
> q:=display([p.s,g.s,q1.s,q2.s,q3.s],scaling=constrained,axes=boxed):
```

```
> T:=T,q
```

```
> od:
```

```
> display(T,insequence=true,scaling=constrained);
```

Epitrocóides

r - raio do círculo móvel

```
> r:=11:
```

R - raio do círculo fixo

```
> R:=18:
```

a - distância do ponto de traçado ao bordo do círculo móvel

se a>0 o ponto de traçado está fora do círculo móvel, se a<0 está dentro e se a=0 o ponto está sobre a circunferência

```
> a:=4:
```

n - número de subdivisão do ângulo percorrido

```
> n:=60:
```

N - número de voltas que o círculo móvel faz

```
> N:=5:
```

Equações:(x,y)=(x(u),y(u)) parametrização da curva a ser traçada

```

> with(plots):
> x:=(R+r)*cos(u)+(r+a)*cos(Pi-(R/r+1)*u):
> y:=(R+r)*sin(u)-(r+a)*sin(Pi-(R/r+1)*u):
> for i from 1 to n do
> te:=2*N*Pi:U:=(te*i)/n:
(c1,c2) são as coordenadas do centro do círculo móvel
> c1:=(R+r)*cos(U):c2:=(R+r)*sin(U):
cp.i : círculo móvel de raio r e centro no ponto de coordenadas (R+r)cos(U) e (R+r)sin(U)
> cp.i:= plot ( [r*cos(t)+c1,r*sin(t)+c2,t=0..2*Pi], scaling = constrained,
color=green):
curva.i : curva traçada pelo ponto (x,y) conforme varia o ângulo U
> curva.i:= plot ( [x,y,u=0..U], scaling = constrained, color=red):
r, s e t são os três segmentos radiais do círculo móvel, indexados por i conforme sua posição durante o movimento
> r.i:= polygonplot ( [[c1,c2], [c1+(r+a)*cos(Pi-(R/r+1)*U),
c2-(r+a)*sin(Pi-(R/r+1)*U)]], scaling = constrained, axes=none,
thickness = 2):
> s.i:= polygonplot ( [[c1,c2], [c1+(r+a)*cos(Pi-(R/r+1)*U+2/3*Pi),
c2-(r+a)*sin(Pi-(R/r+1)*U+2/3*Pi)]], scaling = constrained, axes = none,
thickness = 2):
> t.i:= polygonplot ( [[c1,c2], [c1+(r+a)*cos(Pi-(R/r+1)*U+4/3*Pi),
c2-(r+a)*sin(Pi-(R/r+1)*U+4/3*Pi)]], scaling = constrained, axes = none,
thickness = 2):
> od:
cg : círculo fixo de raio R e centro na origem
> cg:= plot ( [R*cos(v),R*sin(v),v=0..2*Pi], scaling = constrained, color
= blue):
rotina para montagem dos n quadros com as diversas posições da curva traçada
> F:=NULL:
> for i from 1 to n do
> q:=display([cp.i,cg,curva.i,r.i,s.i,t.i],scaling=constrained,axes=boxed):
> F:=F,q
> od:
> display(F,insequence=true,scaling=constrained);

```

```
*****
```

Ciclóides (exercício: inclua comentários passo-a-passo)

```
*****
```

```
> with(plots):
> r:=2:
> a:=0.5:
> n:=30:
> N:=3:
> for i from 1 to n do
> te:=2*N*Pi:U:=(te*i)/n:
> cent:=[U*r,r]:
> bordo1:=[r*U-(r+a)*sin(U),r-(r+a)*cos(U)]:
> bordo2:=[r*U-(r+a)*sin(U+2/3*Pi),r-(r+a)*cos(U+2/3*Pi)]:
> bordo3:=[r*U-(r+a)*sin(U+4/3*Pi),r-(r+a)*cos(U+4/3*Pi)]:
> p.i:= plot ( [r*t-(r+a)*sin(t),r-(r+a)*cos(t), t=0..U], scaling =
constrained, color = blue):
> curva.i:= plot ( [r*cos(u)+U*r,r*sin(u)+r,u=0..2*Pi], scaling =
constrained, color=red):
> q1.i:= polygonplot ( [cent,bordo1]):
> q2.i:= polygonplot ( [cent,bordo2]):
> q3.i:= polygonplot ( [cent,bordo3]):
> q4.i:= polygonplot ( [[0,0],[r*te,0]]):
> od:
> T:=NULL:
> for s from 1 to n do
> q:= display ( [p.s,curva.s,q1.s,q2.s,q3.s,q4.s], axes = boxed):
> T:= T,q
> od:
> display(T,insequence=true,scaling=constrained);
```